
[bookmark: _Toc85286600][bookmark: _Toc87780894][bookmark: _Toc74248357]Ficha técnica
[bookmark: _Toc74248358]Contexto da actividade
	Módulo
	Duración
	Unidade didáctica.
	Sesións 60´
	Actividades
	Sesións 60´

	MP0944. Simulación de sistemas mecatrónicos
	70
	UD01. Deseño de prototipos mecatrónicos con SolidWorks.
	15
	
	

	
	
	UD02. Simulación e validación de sistemas mecatrónicos con SolidWorks.
	15
	
	

	
	
	
	
	
	

	
	
	UD03. Simulación do funcionamento de células robotizadas con ABB RobotStudio.
	20
	
	

	
	
	UD04. Integración de sistemas de adquisición de datos (SAD). Visión artificial con Arduino.
	15
	A01. Proceso de adquisición de datos
	0,5

	
	
	
	
	A02. Esquema de bloques dun SAD, transdutores, convertedores e acondicionamento do sinal.
	1,0

	
	
	
	
	A03. Adquisición de datos mediante visión artificial.
	1,0

	
	
	
	
	A04. Elementos e funcionamento dos sistemas de visión artificial: robot ABB IRB 120 e Arduino. 
	0,5

	
	
	
	
	A05. Programación, montaxes e visión artificial con Arduino.
	10,0

	
	
	
	
	A06. Procedementos avanzados de visión artificial con Arduino
	2,0

	
	
	UD05. Simulación de procesos mecatrónicos complexos con FactoryIO.
	5
	
	


NOTA: Esta actividade está vinculada á programación recollida no arquivo CSIMA03_MP0944_UD04_A05_Programacion_montaxes_vision_artificial_Arduino.pdf
[bookmark: _Toc74248359]Título da actividade
	Nº
	Título
	Descrición
	Duración

	A05
	Programación, montaxes e visión artificial con Arduino.
	Programación e montaxe de Arduinos. Utilización de Arduino para o recoñecemento de datos mediante visión artificial.
	10


[bookmark: _Toc258834878][bookmark: _Toc85286602][bookmark: _Toc87780896][bookmark: _Toc74248360]Resultados de aprendizaxe do currículo
	Resultados de aprendizaxe do currículo
	Completo

	RA3 - Integra sistemas de adquisición de datos en contornos de simulación, monitorizando o estado do sistema mecatrónico e verificando o seu funcionamento
	Non


[bookmark: _Toc74248361]Obxectivos didácticos e título e descrición da actividade
	Obxectivos específicos
	Actividade
	Descrición básica
	Duración

	O5.1
	Programar Arduinos para recoñecer escenas.
	A05
	Programación, montaxes e visión artificial con Arduino.
	Programación e montaxe de Arduinos. Utilización de Arduino para o recoñecemento de datos mediante visión artificial.
	10

	O5.2
	Montar Arduinos para recoñecer escenas.
	
	
	
	

	O5.3
	Empregar Arduinos para monitorizar o estado do sistema.
	
	
	
	

	O5.4
	Verificar o funcionamento da visión artificial con Arduinos.
	
	
	
	


[bookmark: _Toc74248362]Criterios de avaliación
	Criterios de avaliación

	CA3.1 Integráronse sistemas de exploración lineal e cámaras de estado sólido.
CA3.2 Aplicáronse as funcións de detección e dixitalización.
CA3.5 Recoñecéronse as escenas.
CA3.6 Monitorizouse o estado do sistema mecatrónico.
CA3.7 Verificouse o funcionamento do sistema mecatrónico.


[bookmark: _Toc272862471][bookmark: _Toc74248363]Contidos
	Contidos

	Recoñecemento de escenas.
Monitorización do estado do sistema.
Verificación do funcionamento.




[bookmark: _Toc258834879]

[bookmark: _Toc273031427][bookmark: _Toc74248364]Actividades de ensino e aprendizaxe e de avaliación, métodos, recursos e instrumentos  de avaliación
	Qué e para qué
	Cómo
	Con qué
	Cómo e con qué se valora
	Duración
(sesións)

	Actividade
(título e descrición)
	Profesorado
(en termos de tarefas)
	Alumnado
(tarefas)
	Resultados 
ou produtos
	Recursos
	Instrumentos e procedementos de avaliación
	

	A05. Programación, montaxes e visión artificial con Arduino.
Programación e montaxe de Arduinos. Utilización de Arduino para o recoñecemento de datos mediante visión artificial.
	Tp5.1 Explicación sobre a plataforma Arduino.
Tp5.2 Explicación e demostración de montaxes con Arduino.
Tp5.3 Explicación e demostración de prácticas de programación con Arduino.
Tp5.4 Explicación do sistema de visión artificial con Arduino e PixyCam: montaxe, programación, recoñecemento de escenas, monitorización e verificación.
	Ta5.1 Atender ás explicacións.
Ta5.2 Realización de montaxes con Arduino.
Ta5.3. Realización de prácticas de programacion con Arduino.
Ta5.4 Realización dun caso práctico de visión artificial con Arduino e PixyCam: montaxe, programación, recoñecemento de escenas, monitorización e verificación.
	Programas con Arduino.
Montaxes con Arduino.
Recoñecemento de escenas con Arduino.

	Ordenador do profesor. 
Ordenadores do alumnado.
Aula virtual.
Software Arduino.
Kits Arduino con Arduinos UNO e MEGA.
Cámaras de visión artificial PixyCam.
Software PixyMon.
	TO.1 - Prácticas con Arduino.
	10






[bookmark: _Toc85286615][bookmark: _Toc87780911][bookmark: _Toc74248365][bookmark: _Toc84218440]A05. Programación, montaxes e visión artificial con Arduino
[bookmark: _Toc74248366]Introdución
A visión artificial é un sistema de adquisición de datos en sistemas mecatrónicos. Os equipos de visión artificial a nivel industrial supoñen costes moi elevados e prohibitivos a nivel educativo. Arduino permite empregar visión artificial mediante cámaras como a PixyCam, con máis limitacións a nivel técnico que as industriais pero moito máis accesibles a nivel económico. Pero para iniciarse neste método de visión artificial primeiro é necesario aprender a programar e facer montaxes con Arduino. 
[bookmark: _Toc74248367]Actividade
[bookmark: _Toc74248368]Que é Arduino?
Arduino é unha plataforma de hardware libre baseada nun microcontrolador con portos de entrada e saída que permite automatizar dende o ordenador.
A linguaxe de programación de Arduino está baseada en C++. Todas as referencias para programar con Arduino atópanse en:
https://www.arduino.cc/reference/en/
O software Arduino pódese descargar aquí:
https://www.arduino.cc/en/software
Toda a información sobre os diferentes hardware de Arduino pódese ver na seguinte ligazón:
https://www.arduino.cc/en/Main/Products
Nas nosas montaxes empregaremos os Arduino UNO e MEGA. Os seus manuais de iniciación atópanse, respectivamente, en:
https://www.arduino.cc/en/Guide/ArduinoUno
https://www.arduino.cc/en/Guide/ArduinoMega2560
As características técnicas e a documentación dos Arduino UNO e MEGA estarán, respectivamente, en:
https://store.arduino.cc/arduino-uno-rev3
https://store.arduino.cc/arduino-mega-2560-rev3
Para facer as montaxes con Arduino empregarase unha protoboard, á que se lle irán conectando os compoñentes necesarios para a automatización contidos en cada kit Arduino. 

[bookmark: _Toc74248369]Kit Arduino:
O kit Arduino do que dispoñemos para as montaxes conta, entre outros, cos seguintes compoñentes:
Arduino UNO.
Arduino MEGA.
Leds de distintas cores.
Resistencias (10 kΩ; 4,7 kΩ; 220Ω).
Botóns Pull Down.
Botóns Pull Up.
Potenciómetro.
Sensor de temperatura e humidade DHT11.
Pantalla LCD-I2C.
Fotoresistencia-LDR.
Relé.
Buzzer.
Sensor de ultrasóns.
Cables de conexión.
Multímetro.
Cámara de visión artificial PixyCam.
[bookmark: _Toc74248370]Que é unha PixyCam?
PixyCam é un sensor de imaxe con driver libre. Pódese conectar a un ordenador, Arduino ou Raspberry Pi, almacenar datos, recoñecer cores, memorizar, rastrexar obxectos…
A páxina oficial de PixyCam é:
https://pixycam.com/
O software de Pixy chámase PixyMon. Este software é capaz de controlar e almacenar os datos da cámara Pixy, permitindo posicionar cores ou obxectos con só marcalos a través do software. PixyMon permite regular o brillo, as cores e almacenar as cores a seguir.
O software da PixyCam pódese descargar na seguinte ligazón:
https://pixycam.com/downloads-pixy2/
A documentación de uso da Pixy pódese consultar aquí:
https://docs.pixycam.com/wiki/doku.php?id=wiki:v2:hooking_up_pixy_to_a_microcontroller_-28like_an_arduino-29
[bookmark: _Toc84218441]
[bookmark: _Toc74248371]
Tarefas
As tarefas propostas son as seguintes:
Tarefa 1. Montaxes con Arduino. Nesta tarefa imos crear na protoboard montaxes con Arduino e cos compoñentes do kit, de dificultade crecente. Todas as prácticas complementarán á anterior, polo que, unha vez realizada a última, a montaxe permitirá o funcionamento desa e de todas as anteriores.
Tarefa 2. Programación con Arduino. Nesta tarefa crearanse os programas que permitirán o funcionamento das montaxes realizadas na tarefa anterior.
[bookmark: _Toc100312613][bookmark: _Toc84218442]Tarefa 3. Caso práctico de visión artificial con Arduino. Nesta tarefa realizarase a montaxe dun Arduino con PixyCam, e un programa para a detección de cores e a súa monitorización.
[bookmark: _Toc74248372]Tarefa 1. Montaxes con Arduino
Iránselle conectando ao Arduino e á protoboard os distintos compoñentes de xeito incremental. Coa montaxe final, na tarefa 2 realizaranse os programas e comprobarase o resultado.
[bookmark: _Toc74248373]Práctica 1
Seis leds:
[image: ]
[bookmark: _Toc74248374]Práctica 2
Botóns Pull Down e Pull Up:
[image: ]
[bookmark: _Toc74248375]Práctica 3
Botón Input Pull Up:
[image: ]
[bookmark: _Toc74248376]Práctica 4
Potenciómetro:
[image: ]
[bookmark: _Toc74248377]Práctica 5
Sensor de temperatura e humidade DHT11 e pantalla LCD-I2C:
[image: ]
[bookmark: _Toc74248378]Práctica 6
Fotoresistencia-LDR:
[image: ]
[bookmark: _Toc74248379]Práctica 7
Relé:
[image: ]
[bookmark: _Toc74248380]Práctica 8
Buzzer:
[image: ]
[bookmark: _Toc74248381]Práctica 9
Sensor de ultrasóns:
Substitúese a placa Arduino UNO, que se empregou nas prácticas anteriores, pola Arduino MEGA, xa que para esta práctica non chega cos pins que ten UNO. 
Empréganse os mesmos pins que se foron empregando en UNO nas montaxes previas e, a maiores, o sensor de ultrasóns conectará o pin Trig ao pin 22 e o pin Echo ao pin 23 da placa Arduino MEGA, ademais da alimentación (Vcc) e a masa (GND), como se amosa na seguinte imaxe:
[image: ]
A montaxe final con todos os compoñentes conectados a Arduino MEGA quedará así:
[image: ]
[bookmark: _Toc74248382]Tarefa 2. Programación con Arduino
Realizarase a programación co software Arduino dos casos prácticos da tarefa 1, e comprobarase o seu resultado.
Nas liñas de comentario de cada práctica iranse describindo as características dos programas e os pasos que se seguiron na súa elaboración. 

[bookmark: _Toc88740940][bookmark: _Toc88970485][bookmark: _Toc88970976][bookmark: _Toc98856243][bookmark: _Toc74248383]Práctica 1
Seis leds:

/*
Acendemos e apagamos 6 leds, usando un array e blucle for
Amosamos por monitor serial o que está facendo o programa

O array almacena os números dos pins

O programa emprega os bucles for para acender 6 LEDs lentamente e logo apagar rapidamente

*/

// Tamaño do array (debe ser unha constante)
const byte tamanoArray = 6;

// Array que contén os números dos pins
const byte arrayPinLed [tamanoArray] = {8, 9, 10, 11, 12, 13};
                             //Indices: 0  1  2   3   4   5   

// Tempo de espera para apagar e acender os LEDs
const int retardoLed = 200;

void setup() {
  //Inicializamos a comunicación serial a 9600 baudios
  Serial.begin(9600);

  // Poñemos os pins en modo OUTPUT
  for (int i = 0; i < tamanoArray; i++) {
    // Modo OUTPUT
    pinMode(arrayPinLed[i], OUTPUT);
    Serial.print("Pin número ");
    Serial.print(i);
    Serial.println(" definido en modo OUTPUT");
  }
}

void loop() {
  // Acendemos os LEDs un a un
  for (int i = 0; i < tamanoArray; i++) {
    // Cambiamos a estado HIGH
    digitalWrite(arrayPinLed[i], HIGH);
    Serial.print("Led número ");
    Serial.print(i);
    Serial.println(" acendido");

    // Retardo
    delay(retardoLed);
  }

  // Apagamos todos os LEDs
  for (int i = 0; i < tamanoArray; i++) {
    // Cambiamos a estado LOW
    digitalWrite(arrayPinLed[i], LOW);
    Serial.print("Led número ");
    Serial.print(i);
    Serial.println(" apagado");
  }
}

[bookmark: _Toc74248384]Práctica 2
Botóns Pull Down e Pull Up:

/*
Ler pulsadores con Arduino
Un pulsador Pull Down (por hardware, que é o único xeito en Arduino), le LOW mentres non se pulsa e HIGH ao pulsar
Un pulsador Pull Up (por hardware, aínda que se podería facer por software tamén), le HIGH mentres non se pulsa e LOW ao pulsar

Referencias: 
https://www.luisllamas.es/leer-un-pulsador-con-arduino/
https://programarfacil.com/blog/utilizar-pulsadores-en-arduino/

*/

// Pins de entrada onde están conectados os pulsadores 
const int pinPulsadorPullDown = 4;
const int pinPulsadorPullUp = 3;

// Función de configuración, só se executa unha vez
void setup() {
  // Configuramos o Monitor Serie cos bits por segundo
  Serial.begin(9600);

  // Poñemos pins en modo entrada INPUT, non é necesario pero quédanos máis claro o uso
  pinMode(pinPulsadorPullDown, INPUT);
  pinMode(pinPulsadorPullUp, INPUT);
}

void loop() {
  // Comprobamos o estado do pin onde está conectado o pulsador Pull Down e asignamos o valor a unha variable
  boolean estadoPulsadorPullDown = digitalRead(pinPulsadorPullDown);
  // Comprobamos o estado do pin onde está conectado o pulsador Pull Up e asignamos o valor a unha variable
  boolean estadoPulsadorPullUp = digitalRead(pinPulsadorPullUp);

  // amosamos datos por monitor serie
  Serial.print("PullDown = ");
  Serial.print(estadoPulsadorPullDown);
  Serial.print(" - PullUp = ");
  Serial.println(estadoPulsadorPullUp);
}

[bookmark: _Toc74248385]Práctica 3
Botón Input Pull Up:

/*
Ler pulsadores con Arduino, engadimos un Pull Up por software
Un pulsador Pull Down por hardware, que é o único xeito en Arduino, le LOW mentres non se pulsa e HIGH ao pulsar
Un pulsador Pull Up por hardware, aínda que se podería facer por software tamén, le HIGH mentres non se pulsa e LOW ao pulsar
Un pulsador Pull Up por software (a resistencia conéctaa Arduino internamente), le HIGH mentres non se pulsa e LOW ao pulsar

Resistencia Pull Up con Arduino por software:
En Arduino podemos aplicar unha resistencia Pull Up nas entradas dixitais a través de código
Simplemente poñendo un pin en modo INPUT_PULLUP coa instrucción pinMode
Resulta moi sinxelo e evitaranos ter que engadir máis elementos aos nosos circuítos

Referencias: 
https://programarfacil.com/blog/arduino-blog/resistencia-pull-up-y-pull-down/
https://www.luisllamas.es/leer-un-pulsador-con-arduino/
https://programarfacil.com/blog/utilizar-pulsadores-en-arduino/

*/

// Pins de entrada onde están conectados os pulsadores 
const int pinPulsadorPullDown = 4;
const int pinPulsadorPullUp = 3;
const int pinPulsadorInputPullUp = 2;

// Función de configuración, só se executa unha vez
void setup() {
  // Configuramos o Monitor Serie cos bits por segundo
  Serial.begin(9600);

  // Poñemos pins en modo entrada INPUT, non é necesario pero quédanos máis claro o uso
  pinMode(pinPulsadorPullDown, INPUT);
  pinMode(pinPulsadorPullUp, INPUT);
  pinMode(pinPulsadorInputPullUp, INPUT_PULLUP);
}

void loop() {
  // Comprobamos o estado do pin onde está conectado o pulsador Pull Down e asignamos o valor a unha variable
  boolean estadoPulsadorPullDown = digitalRead(pinPulsadorPullDown);
  // Comprobamos o estado do pin onde está conectado o pulsador Pull Up e asignamos o valor a unha variable
  boolean estadoPulsadorPullUp = digitalRead(pinPulsadorPullUp);
  // Comprobamos o estado do pin onde está conectado o pulsador INPU5 Pull Up e asignamos o valor a unha variable
  boolean estadoPulsadorInputPullUp = digitalRead(pinPulsadorInputPullUp);

  // amosamos datos por monitor serie
  Serial.print("PullDown = ");
  Serial.print(estadoPulsadorPullDown);
  Serial.print(" - PullUp = ");
  Serial.print(estadoPulsadorPullUp);
  Serial.print(" - InputPullUp = ");
  Serial.println(estadoPulsadorInputPullUp);
}

[bookmark: _Toc74248386]Práctica 4
Potenciómetro:

[bookmark: _Toc74248387]Práctica 4.1

/*
Lemos a saída dun potenciómetro nun pin analóxico

Exemplo para demostrar o uso do ADC de Arduino (Conversor Analóxico Dixital)

Referencias: 
https://www.arduino.cc/reference/en/language/functions/analog-io/analogread/
https://programarfacil.com/blog/el-potenciometro-y-arduino/
https://www.luisllamas.es/lectura-de-un-potenciometro-con-arduino/
https://programarfacil.com/blog/divisor-de-tension-en-arduino-multiplica-tus-entradas-digitales/

*/

// O pin analóxico onde temos conectado o sensor
byte pinSensor = A0;

// Función de configuración, só se executa unha vez
void setup() {

  // Configuramos o Monitor Serie cos bits por segundo
  Serial.begin(9600);

}

// Función que se repite indefinidamente
void loop() {
  // Obtemos o valor do pin analóxico e almacenámolo na variable
  int valorSensor = analogRead(pinSensor);

  // Enviamos a información ao porto serie
  Serial.print("Valor sensor = ");
  Serial.println(valorSensor);

  /* 
  // Mapeo do valor do sensor noutro rango
  // utilizando a función map()
  int novoValorSensor = map(valorSensor, 0, 1023, 0, 500);
  float valorVoltios=(float)novoValorSensor/100;
  //Amosamos a información no monitor serie
  Serial.print("Valor sensor mapeado a Voltios = ");
  Serial.println(valorVoltios);

  // un pouco de retardo para ler con comodidade
  delay(500);
  */  

}

[bookmark: _Toc74248388]Práctica 4.2

/*
Lemos a saída dun potenciómetro nun pin analóxico
e aplicando a función map e a estrutura case, acendemos a tira de leds

Exercicio aplicación switch case e map()
*/

// Pin analóxico do potenciómetro
const byte pinSensor = A0;

// Pins dixitais LEDs
const byte pinLed1 = 8;
const byte pinLed2 = 9;
const byte pinLed3 = 10;
const byte pinLed4 = 11;
const byte pinLed5 = 12;
const byte pinLed6 = 13;

// Variables de axuda para percorrer bucles for
const byte primePinLed = 8;
const byte ultimoPinLed = 13;

void setup() {
  // Configuración comunicación serie
  Serial.begin(9600);

  // Establecemos o modo dos pins cun bucle
  for (int i = primePinLed; i <= ultimoPinLed; i++) {
    pinMode(i, OUTPUT);
  }
}

void loop() {
  // Ler o pin analóxico do potenciómetro
  int valorSensor = analogRead(pinSensor);

  // Mapeamos o resultado do rango 0 a 1023 en 0 a 6
  byte mapSensor = map(valorSensor, 0, 1023, 0, 6);

  Serial.print("Valor sensor: ");
  Serial.print(valorSensor);
  Serial.print("  Valor sensor map: ");
  Serial.println(mapSensor);

  // Dependendo do valor obtido da función map()
  // executamos un caso ou otro
  switch (mapSensor) {
    case 0:
      // Apagamos todos os LEDs
      for (int i = primePinLed; i <= ultimoPinLed; i++) {
        digitalWrite(i, LOW);
      }
      // Acendemos o LED 1
      digitalWrite(pinLed1, HIGH);
      //Saímos do caso e do switch case
      break;
    case 1:
      // Apagamos todos os LEDs
      for (int i = primePinLed; i <= ultimoPinLed; i++) {
        digitalWrite(i, LOW);
      }
      // Acendemos o LED 2
      digitalWrite(pinLed2, HIGH);
      //Saímos do caso e do switch case
      break;
    case 2:
      // Apagamos todos os LEDs
      for (int i = primePinLed; i <= ultimoPinLed; i++) {
        digitalWrite(i, LOW);
      }
      // Acendemos o LED 3
      digitalWrite(pinLed3, HIGH);
      //Saímos do caso e do switch case
      break;
    case 3:
      // Apagamos todos os LEDs
      for (int i = primePinLed; i <= ultimoPinLed; i++) {
        digitalWrite(i, LOW);
      }
      // Acendemos o LED 4
      digitalWrite(pinLed4, HIGH);
      //Saímos do caso e do switch case
      break;
    case 4:
      // Apagamos todos os LEDs
      for (int i = primePinLed; i <= ultimoPinLed; i++) {
        digitalWrite(i, LOW);
      }
      // Acendemos o LED 5
      digitalWrite(pinLed5, HIGH);
      //Saímos do caso e do switch case
      break;
    case 5:
      // Apagamos todos os LEDs
      for (int i = primePinLed; i <= ultimoPinLed; i++) {
        digitalWrite(i, LOW);
      }
      // Acendemos o LED 6
      digitalWrite(pinLed6, HIGH);
      //Saímos do caso e do switch case
      break;
    case 6:
      // Acendemos todos os LEDs
      for (int i = primePinLed; i <= ultimoPinLed; i++) {
        digitalWrite(i, HIGH);
      }
      //Saímos do caso e do switch case
      break;
  }
}

[bookmark: _Toc74248389]Práctica 5
Sensor de temperatura e humidade DHT11 e pantalla LCD-I2C:

[bookmark: _Toc74248390]Práctica 5.1

/*

É necesaria a montaxe dun sensor de temperatura DHT11

Referencia: https://www.luisllamas.es/arduino-dht11-dht22/

Dirección librerías:
https://github.com/adafruit/DHT-sensor-library
https://github.com/adafruit/Adafruit_Sensor
*/

#include "DHT.h"
 
// Uncomment whatever type you're using!
#define DHTTYPE DHT11   // DHT 11
//#define DHTTYPE DHT22   // DHT 22  (AM2302), AM2321
//#define DHTTYPE DHT21   // DHT 21 (AM2301)
 
// Connect pin 1 (on the left) of the sensor to +5V
// NOTE: If using a board with 3.3V logic like an Arduino Due connect pin 1
// to 3.3V instead of 5V!
// Connect pin 2 of the sensor to whatever your DHTPIN is
// Connect pin 4 (on the right) of the sensor to GROUND
// Connect a 10K resistor from pin 2 (data) to pin 1 (power) of the sensor
 
const int DHTPin = 5;     // what digital pin we're connected to
 
DHT dht(DHTPin, DHTTYPE);
 
void setup() {
   Serial.begin(9600);
   Serial.println("DHTxx test!");
 
   dht.begin();
}
 
void loop() {
   // Wait a few seconds between measurements.
   delay(2000);
 
   // Reading temperature or humidity takes about 250 milliseconds!
   float h = dht.readHumidity();
   float t = dht.readTemperature();
 
   if (isnan(h) || isnan(t)) {
      Serial.println("Erro lendo o sensor DHT!");
      return;
   }
 
   Serial.print("Humidade: ");
   Serial.print(h);
   Serial.print(" %\t");
   Serial.print("Temperatura: ");
   Serial.print(t);
   Serial.println(" *C ");
}

[bookmark: _Toc74248391]Práctica 5.2

/*

É necesaria a montaxe do sensor de temperatura e o LCD

Trátase de combinar os dous proxectos anteriores, 
de maneira que amosamos a temperatura e humidade do sensor DHT11
na pantalla LCD

*/

//librería para sensores DHT
#include "DHT.h"

// Uncomment whatever type you're using!
#define DHTTYPE DHT11   // DHT 11
//#define DHTTYPE DHT22   // DHT 22  (AM2302), AM2321
//#define DHTTYPE DHT21   // DHT 21 (AM2301)

// Connect pin 1 (on the left) of the sensor to +5V
// NOTE: If using a board with 3.3V logic like an Arduino Due connect pin 1
// to 3.3V instead of 5V!
// Connect pin 2 of the sensor to whatever your DHTPIN is
// Connect pin 4 (on the right) of the sensor to GROUND
// Connect a 10K resistor from pin 2 (data) to pin 1 (power) of the sensor

const int DHTPin = 5;     // what digital pin we're connected to

//declarar obxecto da clase DHT:
DHT dht(DHTPin, DHTTYPE);

// librería para LCD I2C
#include <LiquidCrystal_I2C.h>

// Declarar obxecto da clase LiquidCrystal_I2C:
LiquidCrystal_I2C lcd(0x27, 16, 2);

void setup() {
   Serial.begin(9600);
   Serial.println("Programa humidade e temperatura en LCD");

   //iniciar sensor
   dht.begin();

   // Iniciar pantalla LCD
   lcd.begin();
   // Apagar luz traseira e pantalla
   lcd.noBacklight();
   lcd.noDisplay();
   delay(1000);
   // Acender luz traseira e pantalla
   lcd.backlight();
   lcd.display();

   // Amosar mensaxe benvida. Comeza nas coordenadas 0,0
   lcd.print("Temp. y Humidade ");
   //         0123456789012345
   //                   111111
   lcd.setCursor(0,1);//posición na primeira columna segunda fila
   lcd.print("  Sensor DHT11  ");
   delay(2000);
   lcd.clear(); //borramos pantalla
   // indicamos cabeceira
   lcd.print("Hum. %  Temp. oC");
   //         0123456789012345
   //                   111111
}
 
void loop() {
   // dous segundos entre lecturas
   delay(1000);
 
   // Reading temperature or humidity takes about 250 milliseconds!
   float h = dht.readHumidity();
   float t = dht.readTemperature();
 
   if (isnan(h) || isnan(t)) {
      Serial.println("Erro lendo o sensor DHT!");
      lcd.setCursor(0,1);
      lcd.print("Erro lectura sensor");
      //         0123456789012345
      return;
   }
 
   Serial.print("Humidade: ");
   Serial.print(h);
   Serial.print(" %\t");
   Serial.print("Temperatura: ");
   Serial.print(t);
   Serial.println(" *C ");

   lcd.setCursor(0,1);
   lcd.print("                    ");
   lcd.setCursor(1,1);
   lcd.print(h);
   lcd.setCursor(9,1);
   lcd.print(t);

}

[bookmark: _Toc74248392]Práctica 6
Fotoresistencia-LDR:

[bookmark: _Toc74248393]Práctica 6.1

/*

É necesaria a montaxe da fotoresistencia e o LCD (pois amosaremos lecturas en LCD)

Trátase de ler o valor da fotoresistencia e amosala no LCD

Referencias: 
   https://programarfacil.com/blog/arduino-blog/resistencia-ldr-buzzer-theremin/
   https://www.luisllamas.es/medir-nivel-luz-con-arduino-y-fotoresistencia-ldr/

*/

// librería para LCD I2C
#include <LiquidCrystal_I2C.h>

// Declarar obxecto da clase LiquidCrystal_I2C:
LiquidCrystal_I2C lcd(0x27, 16, 2);
// Constantes dos compoñentes conectados aos pins

#define LDR A1

void setup() {

   Serial.begin(9600);
   Serial.println("Lectura de Fotoresistencia ou LDR");

       // Iniciar pantalla LCD
   lcd.begin();
   // Apagar luz traseira e pantalla
   lcd.noBacklight();
   lcd.noDisplay();
   delay(1000);
   // Acender luz traseira e pantalla
   lcd.backlight();
   lcd.display();

   // Amosar mensaxe benvida. Comenza nas coordenadas 0,0
   lcd.print("  Lectura LDR   ");
   //         0123456789012345
   //                   111111
   lcd.setCursor(0,1);//posición en primeira columna segunda fila
   lcd.print("FOTORRESISTENCIA");
   delay(2000);
   lcd.clear(); //borramos pantalla
   // indicamos cabeceira
   lcd.print("Valor LDR 0-1023");
   //         0123456789012345
   //                   111111
}
 
void loop() {
   // dous segundos entre lecturas
   delay(1000);
   // Obtemos o valor da entrada analóxica de 0 a 1023
   int valor = analogRead(LDR);
   Serial.println(valor);

   lcd.setCursor(0,1);
   lcd.print("                    ");
   lcd.setCursor(0,1);
   lcd.print(valor);

}


[bookmark: _Toc74248394]Práctica 6.2

/*

É necesaria a montaxe da fotoresistencia, os 6 leds e o LCD (pois amosaremos lecturas no LCD)

Trátase de ler o valor da fotoresistencia, amosala no LCD 
e acender a tira de leds en función do valor obtido, canta máis luz, máis leds acendemos, de 1 a 6

Referencias: 
   https://programarfacil.com/blog/arduino-blog/resistencia-ldr-buzzer-theremin/
   https://www.luisllamas.es/medir-nivel-luz-con-arduino-y-fotoresistencia-ldr/

*/

// Librería para LCD I2C
#include <LiquidCrystal_I2C.h>

// Declarar obxecto da clase LiquidCrystal_I2C:
LiquidCrystal_I2C lcd(0x27, 16, 2);
// Constantes dos compoñentes conectados aos pins

#define LDR A1

// Tamaño do array (debe ser unha constante)
const byte tamanoArray = 6;
// Array que contén os números dos pins
const byte arrayPinLed [tamanoArray] = {8, 9, 10, 11, 12, 13};
                             //Índices: 0  1  2   3   4   5   

void setup() {

   Serial.begin(9600);
   Serial.println("Lectura de Fotoresistencia ou LDR");

   // Poñemos os pins en modo OUTPUT
   for (int i = 0; i < tamanoArray; i++) {
       // Modo OUTPUT
       pinMode(arrayPinLed[i], OUTPUT);
       Serial.print("Pin número ");
       Serial.print(i);
       Serial.println(" definido en modo OUTPUT");
   }

      // Iniciar pantalla LCD
   lcd.begin();
   // Apagar luz traseira e pantalla
   lcd.noBacklight();
   lcd.noDisplay();
   delay(1000);
   // Acender luz traseira e pantalla
   lcd.backlight();
   lcd.display();

   // Amosar mensaxe benvida. Comenza nas coordenadas 0,0
   lcd.print("  Lectura LDR   ");
   //         0123456789012345
   //                   111111
   lcd.setCursor(0,1);//posición en primeira columna segunda fila
   lcd.print("FOTORESISTENCIA");
   delay(2000);
   lcd.clear(); //borramos pantalla
   // indicamos cabeceira
   lcd.print("Valor LDR 0-1023");
   //         0123456789012345
   //                   111111
}
 
void loop() {
   // dous segundos entre lecturas
   delay(1000);
   // Obtemos o valor da entrada analóxica de 0 a 1023
   int valor = analogRead(LDR);
   Serial.println(valor);

   // Mapeamos o resultado do rango 0 a 1023 en -1 a 5
   int mapValor = map(valor, 0, 1023, -1, 5);

   for(int i=0; i<=mapValor; i++){
      digitalWrite(arrayPinLed[i], HIGH);
      Serial.print("acendendo led "); Serial.println(i);
   }
   for(int i=mapValor+1; i<tamanoArray; i++){
      digitalWrite(arrayPinLed[i], LOW);
      Serial.print("apagando led "); Serial.println(i);
   }

   lcd.setCursor(0,1);
   lcd.print("                    ");
   lcd.setCursor(0,1);
   lcd.print(valor);
   lcd.print(" -> map: ");
   lcd.print(mapValor);

}

[bookmark: _Toc74248395]Práctica 7
Relé:

/*

É necesaria a montaxe do relé, a fotoresistencia, os 6 leds e o LCD (pois amosaremos lecturas no LCD)

Trátase de ler o valor da fotoresistencia, amosala no LCD 
, acender a tira de leds en función do valor obtido, canta máis luz, máis leds acendemos, de 1 a 6
e activar relé por debaixo dun certo nivel de luz .

Referencias: 
   https://www.luisllamas.es/arduino-salida-rele/
   https://programarfacil.com/blog/arduino-blog/resistencia-ldr-buzzer-theremin/
   https://www.luisllamas.es/medir-nivel-luz-con-arduino-y-fotoresistencia-ldr/
*/

// Librería para LCD I2C
#include <LiquidCrystal_I2C.h>

// Declarar obxecto da clase LiquidCrystal_I2C:
LiquidCrystal_I2C lcd(0x27, 16, 2);
// Constantes dos compoñentes conectados aos pins

#define RELE 7
#define LDR A1

// Tamaño do array (debe ser unha constante)
const byte tamanoArray = 6;
// Array que contén os números dos pins
const byte arrayPinLed [tamanoArray] = {8, 9, 10, 11, 12, 13};
                             //Indices: 0  1  2   3   4   5   

void setup() {

   Serial.begin(9600);
   Serial.println("Lectura de Fotoresistencia ou LDR");

   //Poñemos relé en modo saída
   pinMode(RELE, OUTPUT);
   digitalWrite(RELE, HIGH);//Relé en repouso
   Serial.print("Pin número ");
   Serial.print(RELE);
   Serial.println(" definido en modo OUTPUT");
   // Poñemos os pins en modo OUTPUT
   for (int i = 0; i < tamanoArray; i++) {
       // Modo OUTPUT
       pinMode(arrayPinLed[i], OUTPUT);
       Serial.print("Pin número ");
       Serial.print(i);
       Serial.println(" definido en modo OUTPUT");
   }

      // Iniciar pantalla LCD
   lcd.begin();
   // Apagar luz traseira e pantalla
   lcd.noBacklight();
   lcd.noDisplay();
   delay(1000);
   // Acender luz traseira e pantalla
   lcd.backlight();
   lcd.display();

   // Amosar mensaxe benvida. Comeza nas coordenadas 0,0
   lcd.print("  Lectura LDR   ");
   //         0123456789012345
   //                   111111
   lcd.setCursor(0,1);//posición en primeira columna segunda fila
   lcd.print("FOTORESISTENCIA");
   delay(2000);
   lcd.clear(); //borramos pantalla
   // indicamos cabeceira
   lcd.print("Valor LDR 0-1023");
   //         0123456789012345
   //                   111111
}
 
void loop() {
   // dous segundos entre lecturas
   delay(1000);
   // Obtemos o valor da entrada analóxica de 0 a 1023
   int valor = analogRead(LDR);

   // Mapeamos o resultado do rango 0 a 1023 en -1 a 5
   int mapValor = map(valor, 0, 1023, -1, 5);

   Serial.print(valor);
   Serial.print(" map -> ");
   Serial.println(mapValor);

   for(int i=0; i<=mapValor; i++){
      digitalWrite(arrayPinLed[i], HIGH);
      Serial.print("acendendo led "); Serial.println(i);
   }
   for(int i=mapValor+1; i<tamanoArray; i++){
      digitalWrite(arrayPinLed[i], LOW);
      Serial.print("apagando led "); Serial.println(i);
   }

   if(mapValor<2) {
      digitalWrite(RELE, LOW);//activamos relé
      Serial.println("Activado relé");
   } else {
      digitalWrite(RELE, HIGH);//desactivamos relé
      Serial.println("Desactivado relé");
   }

   lcd.setCursor(0,1);
   lcd.print("                    ");
   lcd.setCursor(0,1);
   lcd.print(valor);
   lcd.print(" -> map: ");
   lcd.print(mapValor);

}

[bookmark: _Toc74248396]Práctica 8
Buzzer:

/*

É necesaria a montaxe do buzzer, do relé, a fotoresistencia, os 6 leds e o LCD (pois amosaremos lecturas no LCD)

Trátase de ler o valor da fotoresistencia, amosala no LCD 
, acender a tira de leds en función do valor obtido, canta máis luz, máis leds acendemos, de 1 a 6
activar relé por debaixo dun certo nivel de luz 
e facer sonar o buzzer (pasivo) en función da luz (máis agudo canta máis luz)

Referencias: 
   https://www.arduino.cc/reference/en/language/functions/advanced-io/tone/
   https://programarfacil.com/blog/arduino-blog/resistencia-ldr-buzzer-theremin/
   https://www.luisllamas.es/arduino-salida-rele/
   https://www.luisllamas.es/medir-nivel-luz-con-arduino-y-fotoresistencia-ldr/
*/

// Librería para LCD I2C
#include <LiquidCrystal_I2C.h>

// Declarar obxecto da clase LiquidCrystal_I2C:
LiquidCrystal_I2C lcd(0x27, 16, 2);
// Constantes dos compoñentes conectados aos pins

#define BUZZER 6 //se o buzzer é pasivo, o pin ten que ser tipo PWM
#define RELE 7
#define LDR A1

// Tamaño do array (debe ser unha constante)
const byte tamanoArray = 6;
// Array que contén os números dos pins
const byte arrayPinLed [tamanoArray] = {8, 9, 10, 11, 12, 13};
                             //Indices: 0  1  2   3   4   5   

void setup() {

   Serial.begin(9600);
   Serial.println("Lectura de Fotoresistencia ou LDR");

   //Poñemos buzzer en modo saída
   pinMode(BUZZER, OUTPUT);
   Serial.print("Pin número ");
   Serial.print(BUZZER);
   Serial.println(" definido en modo OUTPUT");

   //Poñemos relé en modo saída
   pinMode(RELE, OUTPUT);
   digitalWrite(RELE, HIGH);//Relé en repouso
   Serial.print("Pin número ");
   Serial.print(RELE);
   Serial.println(" definido en modo OUTPUT");
   // Poñemos os pins en modo OUTPUT
   for (int i = 0; i < tamanoArray; i++) {
       // Modo OUTPUT
       pinMode(arrayPinLed[i], OUTPUT);
       Serial.print("Pin número ");
       Serial.print(i);
       Serial.println(" definido en modo OUTPUT");
   }

      // Iniciar pantalla LCD
   lcd.begin();
   // Apagar luz traseira e pantalla
   lcd.noBacklight();
   lcd.noDisplay();
   delay(1000);
   // Acender luz traseira e pantalla
   lcd.backlight();
   lcd.display();

   // Amosar mensaxe benvida. Comenza nas coordenadas 0,0
   lcd.print("  Lectura LDR   ");
   //         0123456789012345
   //                   111111
   lcd.setCursor(0,1);//posición en primeira columna segunda fila
   lcd.print("FOTORESISTENCIA");
   delay(2000);
   lcd.clear(); //borramos pantalla
   // indicamos cabeceira
   lcd.print("Valor LDR 0-1023");
   //         0123456789012345
   //                   111111
}
 
void loop() {
   // dous segundos entre lecturas
   delay(1000);
   // Obtemos o valor da entrada analóxica de 0 a 1023
   int valor = analogRead(LDR);

   // Mapeamos o resultado do rango 0 a 1023 en -1 a 5
   int mapValor = map(valor, 0, 1023, -1, 5);

   // Obtemos a frecuencia en función do voltaxe que entra
   // Este voltaxe dependerá da resistencia LDR e da luz que incida sobre ela
   int frecuencia = 400 + (valor / 2);

   // Empregamos a función tone para reproducir o son no pin onde
   // teñamos conectado o buzzer e a frecuencia que queiramos
   tone(BUZZER, frecuencia, 100);

   Serial.print(valor);
   Serial.print(" map -> ");
   Serial.println(mapValor);

   for(int i=0; i<=mapValor; i++){
      digitalWrite(arrayPinLed[i], HIGH);
      Serial.print("acendendo led "); Serial.println(i);
   }
   for(int i=mapValor+1; i<tamanoArray; i++){
      digitalWrite(arrayPinLed[i], LOW);
      Serial.print("apagando led "); Serial.println(i);
   }

   if(mapValor<2) {
      digitalWrite(RELE, LOW);//activamos relé
      Serial.println("Activado relé");
   } else {
      digitalWrite(RELE, HIGH);//desactivamos relé
      Serial.println("Desactivado relé");
   }

   lcd.setCursor(0,1);
   lcd.print("                    ");
   lcd.setCursor(0,1);
   lcd.print(valor);
   lcd.print(" -> map: ");
   lcd.print(mapValor);

}

[bookmark: _Toc74248397]Práctica 9
Sensor de ultrasóns:

/*

Vindo do esquema anterior, se se quere manter todo o anterior, temos que pasar a Arduino MEGA

É necesaria a montaxe do ultrasóns, buzzer, os 6 leds e o LCD (pois amosaremos lecturas no LCD)

Trátase de ler a distancia, amosala no LCD 
, acender leds verdes, amarelos ou vermellos en función da distancia (3 rangos de distancia)
e facer soar o buzzer (pasivo) en función da distancia, máis agudo canto máis preto (3 rangos)

Referencias: 
   https://programarfacil.com/blog/arduino-blog/sensor-ultrasonico-arduino-medir-distancia/
   https://www.luisllamas.es/medir-distancia-con-arduino-y-sensor-de-ultrasonidos-hc-sr04/
*/

// Librería para LCD I2C
#include <LiquidCrystal_I2C.h>

// Declarar obxecto da clase LiquidCrystal_I2C:
LiquidCrystal_I2C lcd(0x27, 16, 2);
// Constantes dos compoñentes conectados aos pins

//pins ultrasóns
#define TRIGGER 22
#define ECHO 23

//pin buzzer
#define BUZZER 6 //se o buzzer é pasivo, o pin ten que de ser tipo PWM

// Tamaño do array (debe ser unha constante)
const byte tamanoArray = 6;
// Array que contén os números dos pins
const byte arrayPinLed [tamanoArray] = {8, 9, 10, 11, 12, 13};
                             //Indices: 0  1  2   3   4   5   

// Constantes para calcular distancia e umbrais
const float son = 34300.0; // Velocidade do son en cm/s
const float umbral1 = 30.0;
const float umbral2 = 20.0;
const float umbral3 = 10.0;

void setup() {

  Serial.begin(9600);
  Serial.println("Lectura de Fotoresistencia ou LDR");

  //Poñemos buzzer en modo saída
  pinMode(BUZZER, OUTPUT);

  // Poñemos os pins led en modo OUTPUT
  for (int i = 0; i < tamanoArray; i++) {
       // Modo OUTPUT
     pinMode(arrayPinLed[i], OUTPUT);
  }

   //modo pins ultrasóns
  pinMode(ECHO, INPUT);
  pinMode(TRIGGER, OUTPUT);   

      // Iniciar pantalla LCD
  lcd.begin();
   // Apagar luz traseira e pantalla
  lcd.noBacklight();
  lcd.noDisplay();
  delay(1000);
   // Acender luz traseira e pantalla
  lcd.backlight();
  lcd.display();

   // Amosar mensaxe benvida. Comeza nas coordenadas 0,0
  lcd.print("  Ultrason   ");
   //         0123456789012345
   //                   111111
   lcd.setCursor(0,1);//posición na primeira columna segunda fila
   lcd.print("     HC-SR04    ");
   delay(2000);
   lcd.clear(); //borramos pantalla
   // indicamos cabeceira
   lcd.print("Distancia en cm:");
   //         0123456789012345
   //                   111111

   // Apagamos todos os LEDs
   apagarLEDs();

}

void loop() {
    delay(500);

     // Preparamos o sensor de ultrasóns
    iniciarTrigger();

     // Obtemos a distancia
    float distancia = calcularDistancia();

    Serial.print(distancia);
    Serial.print("cm");
    Serial.println();

    lcd.setCursor(0,1);
    lcd.print("                ");
    lcd.setCursor(0,1);
    lcd.print(distancia);
    lcd.print(" cm");

     // Apagamos todos os LEDs
    apagarLEDs();

     // Lanzamos alerta se estamos dentro do rango de perigo
    if (distancia < umbral1)
    {
       // Lanzamos alertas
       alertas(distancia);
   }

}

// Apaga todos os LEDs
void apagarLEDs()
{
     // Apagamos todos os LEDs
    for(byte i=0; i<tamanoArray; i++){
       digitalWrite(arrayPinLed[i], LOW);
    } 

}

void acenderLEDsDendeAta(byte dende, byte ata) {
    for(byte i=dende; i<=ata; i++){
       digitalWrite(arrayPinLed[i], HIGH);
    } 

}

// Función que comproba se hai que lanzar algunha alerta visual ou sonora
void alertas(float distancia) {
       if (distancia < umbral1 && distancia >= umbral2)
       {
          // Acendemos LEDs verdes
        acenderLEDsDendeAta(0,1);
        tone(BUZZER, 2000, 200);
      }
      else if (distancia < umbral2 && distancia > umbral3)
      {
          // Acendemos LEDs amarelos
        acenderLEDsDendeAta(2,3);
        tone(BUZZER, 2500, 200);
      }
      else if (distancia <= umbral3)
      {
          // Acendemos LEDs vermellos
        acenderLEDsDendeAta(4,5);
        tone(BUZZER, 3000, 200);
      }
}

// Método que calcula a distancia á que se atopa un obxecto
// Devolve unha variable tipo float que contén a distancia
float calcularDistancia()
{
        // A función pulseIn obtén o tempo que tarda en cambiar entre estados, neste caso a HIGH
       unsigned long tempo = pulseIn(ECHO, HIGH);

        // Obtemos a distancia en cm, hai que convertir o tempo en segundos xa que está en microsegundos
        // por iso se multiplica por 0.000001
       float distancia = tempo * 0.000001 * son / 2.0;
       return distancia;
}

// Método que inicia a secuencia do Trigger para comezar a medir
void iniciarTrigger()
{
        // Poñemos o Trigger en estado baixo e agardamos 2 ms
       digitalWrite(TRIGGER, LOW);
       delayMicroseconds(2);

        // Poñemos o pin Trigger en estado alto e agardamos 10 ms
       digitalWrite(TRIGGER, HIGH);
       delayMicroseconds(10);

        // e poñemos o pin Trigger en estado baixo
       digitalWrite(TRIGGER, LOW);
}

[bookmark: _Toc74248398]Tarefa 3. Caso práctico de visión artificial con Arduino

[bookmark: _Toc74248399]Práctica 10
Caso práctico de identificación de cores con Arduino e PixyCam.
[bookmark: _Toc84218449]
//
// begin license header
//
// This file is part of Pixy CMUcam5 or "Pixy" for short
//
// All Pixy source code is provided under the terms of the
// GNU General Public License v2 (http://www.gnu.org/licenses/gpl-2.0.html).
// Those wishing to use Pixy source code, software and/or
// technologies under different licensing terms should contact us at
// cmucam@cs.cmu.edu. Such licensing terms are available for
// all portions of the Pixy codebase presented here.
//
// end license header
//
// This sketch is a good place to start if you're just getting started with 
// Pixy and Arduino.  This program simply prints the detected object blocks 
// (including color codes) through the serial console.  It uses the Arduino's 
// ICSP SPI port.  For more information go here:
//
// https://docs.pixycam.com/wiki/doku.php?id=wiki:v2:hooking_up_pixy_to_a_microcontroller_-28like_an_arduino-29
//
  
#include <Pixy2.h>

// This is the main Pixy object 
Pixy2 pixy;

const char* nombreObjeto[] = {
    "rojo",
    "amarillo",
    "violeta",
    "verde",
    "rosa",
    "verde claro",
    "turquesa"
};

byte nObjeto;

void setup()
{
  Serial.begin(115200);
  Serial.print("Starting...\n");
  
  pixy.init();
}

void loop()
{ 
  int i; 
  // grab blocks!
  pixy.ccc.getBlocks();
  
  // If there are detect blocks, print them!
  if (pixy.ccc.numBlocks)
  {
    Serial.print("Detectados ");
    Serial.print(pixy.ccc.numBlocks);
    Serial.println(" objetos:");
    for (i=0; i<pixy.ccc.numBlocks; i++)
    {
      Serial.print("  Objeto ");
      Serial.print(i);
      Serial.print(": ");
      nObjeto=pixy.ccc.blocks[i].m_signature;
      Serial.println(nombreObjeto[nObjeto - 1]);
      //pixy.ccc.blocks[i].print();
    }
  } else {
    Serial.println("-- no se detectan objetos ---"); 
  }  
}

Amosaranse á PixyCam artigos de distintas cores e gardaranse estas cores cos seus nomes correspondentes, os mesmos que figuran no programa de Arduino anterior:

Posteriormente, ao presentárselle á PixyCam obxectos das cores gardadas, detectaráas e indicaráas no Monitor Serie:

[bookmark: _Toc74248400]Materiais
[bookmark: _Toc84218450][bookmark: _Toc74248401]Textos de apoio ou de referencia
https://www.arduino.cc/
https://www.luisllamas.es/tutoriales-de-arduino/
https://programarfacil.com/blog/arduino-blog/arduino-uno-r3/
https://store.prometec.net/tutoriales/
https://www.tinkercad.com/
https://pixycam.com/
[bookmark: _Toc84218451][bookmark: _Toc74248402]Recursos didácticos
[bookmark: _Toc84218452]Ordenador do profesor.
Ordenadores do alumnado.
Aula virtual.
Software Arduino.
Kits Arduino con Arduinos UNO e MEGA.
Cámaras de visión artificial PixyCam.
Software PixyMon.


[bookmark: _Toc42332273][bookmark: _Toc74248403]Avaliación
	Criterios de avaliación seleccionados para esta actividade
	Instrumento de avaliación

	· CA3.1 Integráronse sistemas de exploración lineal e cámaras de estado sólido.
	Táboa de observación.

	· CA3.2 Aplicáronse as funcións de detección e dixitalización.
	Táboa de observación.

	· CA3.5 Recoñecéronse as escenas.
	Táboa de observación. 

	CA3.6 Monitorizouse o estado do sistema mecatrónico.
	Táboa de observación.

	CA3.7 Verificouse o funcionamento do sistema mecatrónico.
	Táboa de observación.



[bookmark: _Toc42332274][bookmark: _Toc74248404]Táboa de observación 
	Rubro
	Habilidade
	Puntos
	Cualificación

	Montaxe
	As montaxes solicitadas foron revisadas e contan con todos os elementos mencionados durante a realización da actividade.
Amósase comprensión profunda ao analizar e recoller os conceptos clave.
	5
	

	
	As montaxes solicitadas foron revisadas e contan con moitos dos elementos mencionados durante a realización da actividade.
Amósanse evidencias de comprensión ao recoller a maioría dos conceptos.
	4
	

	
	As montaxes solicitadas presentáronse sen revisar e contan cos elementos básicos mencionados durante a realización da actividade.
	3
	

	
	Só parte das montaxes solicitadas contan cos elementos mencionados durante a realización da actividade.
	2
	

	
	O contido das montaxes é deficiente.
	1
	

	
	Non se realizan as montaxes.
	0
	

	Programación
	Os programas solicitados foron revisados e contan con todas as características mencionadas durante a realización da actividade.
Amósase comprensión profunda do proceso e do resultado.
	5
	

	
	Os programas solicitados foron revisados e contan con moitas das características mencionadas durante a realización da actividade.
Amósanse evidencias de comprensión do proceso e do resultado.
	4
	

	
	Os programas solicitados presentáronse sen revisar e contan coas características básicas mencionados durante a realización da actividade.
	3
	

	
	Só parte dos programas solicitados contan coas características mencionadas durante a realización da actividade.
	2
	

	
	O contido dos programas é deficiente.
	1
	

	
	Non se realizan os programas.
	0

